```
$Version
```

```
10.0 for Linux x86 (64-bit) (September 9, 2014)
```

```
Off[General::spell, General::spell1]
```

INDEX:

How small a fraction can you use to approximate Pi?
Express the Golden Ratio as a continued fraction:
Expand a Polynomial:
Solve a cubic equation:
Solve a quartic equation:
Solve a fifth order equation:
Do some integrations:
Do some plots:
Do some 3D Plots:
Do some Contour Plots:
Do some Series
Do some Matrices
Do some 3D Plots
Do some Bessel Functions
How to View Animations in Mathematica
Standing Wave Animation
Play with Loops and Conditionals:
3D Plots of Complex Functions
Prime Numbers
The Factorial Function: Old Fashioned Way
The Factorial Function: New Way
Different Bases
Least Squares Fitting:
Musical Beats
Solve and Reduce
Fractals

How small a fraction can you use to approximate Pi?

```
N[Pi,100]
```

```
3.14159265358979323846264338327950288419716939937510582097494459230781640628620899
 86280348253421170 68
```

```
Timing[N[Pi,1000]][[1]]
```

```
0.000227
```

```
Do[out=Rationalize[N[Pi],10^(-m)];
   Print[m," = ",out,"   ",N[out,10]];Print["  "],{m,1,8}]
```

$1 = \frac{22}{7}$ 3.142857143

$2 = \frac{22}{7}$ 3.142857143

$3 = \frac{201}{64}$ 3.140625000

$4 = \frac{333}{106}$ 3.141509434

$5 = \frac{355}{113}$ 3.141592920

$6 = \frac{355}{113}$ 3.141592920

$7 = \frac{75\,948}{24\,175}$ 3.141592554

$8 = \frac{100\,798}{32\,085}$ 3.141592645

Express the Golden Ratio as a continued fraction:

```
Nest[Function[x,1/(1+x)],x,3]
```

$$\frac{1}{1+\frac{1}{1+\frac{1}{1+x}}}$$

```
goldenratio=1+Nest[Function[x,1/(1+x)],x,10]
```

$$1+\cfrac{1}{1+\cfrac{1}{1+\cfrac{1}{1+\cfrac{1}{1+\cfrac{1}{1+\cfrac{1}{1+\cfrac{1}{1+\cfrac{1}{1+\cfrac{1}{1+\cfrac{1}{1+x}}}}}}}}}}}$$

```
N[goldenratio//.{x->1},20]
```

```
1.6180555555555555556
```

```
N[GoldenRatio,20]
```

```
1.6180339887498948482
```

Expanda Polynomial:

```
temp=Expand[(a+b)^20]
```

$$a^{20}+20\,a^{19}\,b+190\,a^{18}\,b^2+1140\,a^{17}\,b^3+4845\,a^{16}\,b^4+15\,504\,a^{15}\,b^5+38\,760\,a^{14}\,b^6+77\,520\,a^{13}\,b^7+125\,970\,a^{12}\,b^8+167\,960\,a^{11}\,b^9+184\,756\,a^{10}\,b^{10}+167\,960\,a^9\,b^{11}+125\,970\,a^8\,b^{12}+77\,520\,a^7\,b^{13}+38\,760\,a^6\,b^{14}+15\,504\,a^5\,b^{15}+4845\,a^4\,b^{16}+1140\,a^3\,b^{17}+190\,a^2\,b^{18}+20\,a\,b^{19}+b^{20}$$

```
Factor[temp]
```

$$(a+b)^{20}$$

Solve a cubicequation:

```
temp=2x^3+3x^2+5x+2
```

$$2+5\,x+3\,x^2+2\,x^3$$

```
sol=Solve[temp==0,x]
```

$$\left\{\left\{x\to -\frac{1}{2}\right\},\left\{x\to \frac{1}{2}\left(-1-i\sqrt{7}\right)\right\},\left\{x\to \frac{1}{2}\left(-1+i\sqrt{7}\right)\right\}\right\}$$

```
N[sol]
```

$$\{\{x\to -0.5\},\{x\to -0.5-1.32288\,i\},\{x\to -0.5+1.32288\,i\}\}$$

Solve a quartic equation:

```
temp=7x^4+2x^3+3x^2+5x+2
```

$$2 + 5\,x + 3\,x^2 + 2\,x^3 + 7\,x^4$$

```
sol=Solve[temp==0,x];
```

```
sol //Short[#,2]&
```

$$\left\{\left\{x \to -\frac{1}{14} + \frac{1}{2}\sqrt{-\frac{13}{49} + \frac{(\text{<<1>>})^{1/3}}{21^{2/3}} + \frac{7^{2/3}}{\left(\frac{3}{2}\,(\text{<<1>>})\right)^{1/3}}} - \frac{1}{2}\sqrt{\left(-\frac{26}{49} - \frac{\left(\frac{1}{2}\left(81 + i\sqrt{\text{<<4>>}\ 1}\right)\right)^{1/3}}{21^{2/3}} - \frac{7^{2/3}}{\left(\frac{3}{2}\,(\text{<<1>>})\right)^{1/3}} - \frac{450}{343\sqrt{-\frac{13}{49} + \frac{(\text{<<1>>})^{1/3}}{21^{2/3}} + \frac{7^{2/3}}{\left(\frac{3}{2}\,(\text{<<1>>})\right)^{1/3}}}}\right)}\right\},\right.$$

$$\{x \to \text{<<1>>}\},\ \{x \to \text{<<1>>}\},\ \left.\left\{x \to -\frac{1}{14} - \frac{1}{2}\sqrt{\text{<<1>>}} + \frac{1}{2}\,\text{<<1>>}\right\}\right\}$$

```
N[sol]
```

```
{{x → 0.380311 - 0.875258 i}, {x → 0.380311 + 0.875258 i},
 {x → -0.523168 - 0.200052 i}, {x → -0.523168 + 0.200052 i}}
```

```
NRoots[temp==0,x]
```

```
x == -0.523168 - 0.200052 i || x == -0.523168 + 0.200052 i ||
 x == 0.380311 - 0.875258 i || x == 0.380311 + 0.875258 i
```

Solve a fifth order equation:

```
temp=17x^4+7x^4+2x^3+3x^2+5x+2
```

$$2 + 5\,x + 3\,x^2 + 2\,x^3 + 24\,x^4$$

```
NRoots[temp==0,x]
```

```
x == -0.37722 - 0.213705 i || x == -0.37722 + 0.213705 i ||
 x == 0.335554 - 0.575108 i || x == 0.335554 + 0.575108 i
```

Do someintegrations:

```
ztemp = 1/((x+I y) -1)^2 //.{x->u+v, y->u-v}
```

$$\frac{1}{(-1+u+i\,(u-v)+v)^2}$$

```
{x->u+v, y->u-v}
```

$$\{x \to u+v,\ y \to u-v\}$$

```
Integrate[ztemp,u]
```

$$-\frac{1-i}{-2+(2+2\,i)\,u+(2-2\,i)\,v}$$

```
Integrate[temp,x]
```

$$2\,x+\frac{5\,x^2}{2}+x^3+\frac{x^4}{2}+\frac{24\,x^5}{5}$$

Do someplots:

```
Plot[2x^3-12x^2+5x+2,{x,-10,10}]
```



```
Plot[Sin[1/x],{x,0,0.1}]
```




```
Plot[Gamma[x],{x,-5,5}]
```



Do some3D Plots:

```
Plot3D[Re[Exp[10 I(x^2+y^2)]],{x,0,1},{y,0,1}]
```



Do someContourPlots:

```
ContourPlot[Re[Exp[10 I(x^2+y^2)]],{x,0,1},{y,0,1}]
```



Do someSeries

```
Series[Exp[x],{x,0,6}]
```

$$1 + x + \frac{x^2}{2} + \frac{x^3}{6} + \frac{x^4}{24} + \frac{x^5}{120} + \frac{x^6}{720} + O[x]^7$$

```
Series[Exp[x],{x,a,4}]
```

$$e^a + e^a (x-a) + \frac{1}{2} e^a (x-a)^2 + \frac{1}{6} e^a (x-a)^3 + \frac{1}{24} e^a (x-a)^4 + O[x-a]^5$$

Do someMatrices

```
matrix={{a,c},{c,b}};
MatrixForm[matrix]
```

$$\begin{pmatrix} a & c \\ c & b \end{pmatrix}$$

```
Det[matrix]
```

$$a\,b - c^2$$

```
Eigenvalues[matrix] //Simplify
```

$$\left\{\frac{1}{2}\left(a + b - \sqrt{a^2 - 2\,a\,b + b^2 + 4\,c^2}\right),\ \frac{1}{2}\left(a + b + \sqrt{a^2 - 2\,a\,b + b^2 + 4\,c^2}\right)\right\}$$

Do some3D Plots

```
res= 1/( (x+I y) -(1+I))
```

$$\frac{1}{(-1 - i) + x + i\,y}$$

```
pic=Plot3D[Re[res],{x,0,2},{y,0,2},
                BoxRatios->{1,1,1},
                ViewPoint->{-2,6,1}]
```




```
Show[pic, ViewPoint->{-2,4,2}, BoxRatios->{1,1,1}]
```



```
Plot[Re[res] /.{y->1} //Evaluate ,{x,0,2}]
```



Do someBesselFunctions

```
list={BesselJ[1,z],BesselJ[2,z]}
```

```
{BesselJ[1, z], BesselJ[2, z]}
```

```
pic2=Plot[{BesselJ[1,z],BesselJ[2,z],BesselJ[3,z],
           BesselJ[4,z],BesselJ[5,z],BesselJ[6,z],
           BesselJ[7,z],BesselJ[8,z],BesselJ[9,z]
            }  ,{z,0,10}]
```



How to View Animations in *Mathematica*

```
Animate[Plot[BesselJ[n, x], {x, 0, 15}], {n, 1, 7, 0.1}]
```



StandingWave Animation

```
Animate[
Plot[
   {Sin[2 Pi ( x - t) ],
    Sin[2 Pi ( x + t) ],
    Sin[2 Pi ( x - t) ]+
    Sin[2 Pi ( x + t) ]},{x,0,1},
PlotStyle->{Thickness[0.005],Thickness[0.005],Thickness[0.008]},
PlotRange->{-2,2} ]
,{t,0,1}]
```



Play with Loops and Conditionals:

```
For[i=1, i<4,  Print[i++]]
```

```
1

2

3
```

```
For[i=1, i<4,  Print[++i]]
```

```
2

3

4
```

```
For[i=2, i<8,  Print[i+=2]]
```

```
4

6

8
```

```
For[i=2, i<64,  Print[i*=2]]
```

```
4

8

16

32

64
```

```
Clear[x,t];
t=x;
Do[t=1/(1+t), {3}];t
```

$$\frac{1}{1+\frac{1}{1+\frac{1}{1+x}}}$$

```
Clear[x,t];
t=x;
table=Table[t=1/(1+t), {4}]
```

$$\left\{\frac{1}{1+x},\ \frac{1}{1+\frac{1}{1+x}},\ \frac{1}{1+\frac{1}{1+\frac{1}{1+x}}},\ \frac{1}{1+\frac{1}{1+\frac{1}{1+\frac{1}{1+x}}}}\right\}$$

```
table //.{x->1} //N
```

```
{0.5, 0.666667, 0.6, 0.625}
```

```
FixedPoint[1/(1+#)&,1.0,20]   //N[#,20]&
```

```
0.618034
```

```
FixedPoint[1/(1+#)&,1.0]    //N[#,20]&
```

```
0.618034
```

```
Nest[ 1/(1+#)&,x,4]
```

$$\frac{1}{1+\frac{1}{1+\frac{1}{1+\frac{1}{1+x}}}}$$

```
t1=Nest[ 1/(1+#)&,1,100] //N[#,20]&
```

```
0.61803398874989484820
```

```
t2=1/Nest[ 1/(1+#)&,1,100] //N[#,20]&
```

```
1.61803398874989484482
```

```
t1-t2 //N[#,20]&
```

```
-1.0000000000000000000
```

3D Plots of Complex Functions

```
Plot3D[Im[Sqrt[x + I y]],{x,-4,4},{y,-4,4}, Lighting->Automatic]
```



```
ContourPlot[Im[Sqrt[x + I y]],{x,-4,4},{y,-4,4}]
```




```
Needs["Graphics`ParametricPlot3D`"]

ParametricPlot3D[
    {r Cos[theta]
    ,r Sin[theta]
    ,Im[Sqrt[r Cos[theta] + I r Sin[theta]]]}
,{r,0,4}
,{theta,-Pi+0.0001, Pi-0.0001}
,Boxed->False
,BoxRatios->{1,1,0.4}]
```



```
ParametricPlot3D[
    {r Cos[theta]
    ,r Sin[theta]
    ,Re[Sqrt[r Cos[theta] + I r Sin[theta]]]}
,{r,0,4}
,{theta,-Pi+0.0001, Pi-0.0001}
,Boxed->False
,BoxRatios->{1,1,0.4}]
```



PrimeNumbers

```
Clear[i];
Do[Print[i,"  ",Prime[i]],{i,1000,2000,100}]
```

```
1000  7919
1100  8831
1200  9733
1300  10 657
1400  11 657
1500  12 553
1600  13 499
1700  14 519
1800  15 401
1900  16 381
2000  17 389
```

```
Plot[Prime[Floor[i]],{i,1,20}];
```

```
Plot[Prime[Floor[i]],{i,1,100}];
```

The FactorialFunction:Old FashionedWay

```
Clear[fac2]
fac2[n_Integer?(Function[n,n>0]) ]:=
 Module[{total=1},
    Do[ total=total * i,{i,1,n}];
    Return[total];
 ]
```

```
fac2[0]=1
```

```
1
```

```
{fac2[4],fac2[0],fac2[4.0],fac2[4.5],fac2[-4]}
```

```
{24, 1, fac2[4.], fac2[4.5], fac2[-4]}
```

The FactorialFunction:New Way

```
Clear[fac]
fac[n_Integer]:= n fac[n-1] /; n>=1
fac[0]=1;
```

```
?fac
```

```
Global`fac
```

```
fac[0] = 1
```

```
fac[n_Integer] := n fac[n - 1] /; n ≥ 1
```

{fac[4],fac[0],fac[4.0],fac[4.5],fac[-4]}

```
{24, 1, fac[4.], fac[4.5], fac[-4]}
```

Log[10,1000.!]

```
2567.604644222133
```

****

DifferentBases

BaseForm[#, 2] & /@ {$MaxMachineNumber, $MaxPrecision}

$\{1._2 \times 2^{1024}, \infty\}$

BaseForm[Pi // N, 2]

11.00100100001111111_2

BaseForm[Pi // N, 16]

$3.243f_{16}$

****

LeastSquaresFitting: b

```
list = {1, 2.5, 3.0, 4.5, 5.0};
fit[x_] = Fit[list, {1, x}, x]
```

```
0.2 + 1. x
```

```
p1 = ListPlot[list, PlotStyle → {PointSize[0.03]}];
```

```
p2 = Plot[fit[x], {x, 1, 5}]; 1
```

```
1
```

```
Show[p1, p2]
```



****

MusicalBeats

```
Plot[Sin[2 Pi 440 t] + Sin[2 Pi 444 t], {t, 0, 1}, PlotPoints → 100]
```




```
f = 440;
Play[Sin[2 Pi 440 t] + Sin[2 Pi 444 t], {t, 0, 1}, SampleRate → 4000]
```



****

****

SolveandReduce1

```
Solve[x^3 == 1, x]
```

$$\left\{\{x \to 1\}, \left\{x \to -(-1)^{1/3}\right\}, \left\{x \to (-1)^{2/3}\right\}\right\}$$

```
Solve[x a == a, x]
```

$$\{\{x \to 1\}\}$$

```
Reduce[x a == a, x]
```

$$a == 0 \mid\mid x == 1$$

****

****

FractalsP

```
Clear[reduce, transx, transy];
reduce[point_] := point / 3;
transx[point_] := point + (2 / 3 {1, 0});
transy[point_] := point + (2 / 3 {0, 1});


p0 = {Random[], Random[]}
list = {};
print = False;

Do[p1 = reduce[p0];
 p1 = If[Random[] > 0.5, transx[p1], p1];
 p1 = If[Random[] > 0.5, transy[p1], p1];
 list = Append[list, p1];
 If[print, Print[p1]];
 p0 = p1;, {i, 1, 3000}]
```

```
{0.866576, 0.320901}
```

```
ListPlot[list, PlotStyle → {PointSize[0.005]},
  PlotRange → {{0, 1}, {0, 1}}, Axes → False, AspectRatio → 1]
```